# Is there a way to plot a single level curve of a function of x and y through a given point?

For example, I am trying to plot the single level curve of

f(x,y)=xsin[y/x]+ysin[2x]


that goes through the point (3pi,3pi). Can anyone show me how to do this?

-

Just for illustrative purposes:

f[x_, y_] := x Sin[y/x] + y Sin[2 x]
pt = f[3 Pi, 3 Pi];
plot = Plot3D[f[x, y], {x, -4 Pi, 4 Pi}, {y, -4 Pi, 4 Pi},
MeshFunctions -> (#3 &), Mesh -> {{pt}}, MeshStyle -> {Thick, Red},
PlotPoints -> 100, ColorFunctionScaling -> False];
Show[plot,
Graphics3D[{Yellow, PointSize[0.03], Point[{{3 Pi, 3 Pi, pt}}]}]]


visualizes the level curve (and point in yellow):

Contour plot can visualize the level curve:

    ContourPlot[f[x, y], {x, -4 Pi, 4 Pi}, {y, -4 Pi, 4 Pi},
Contours -> {pt}, ContourShading -> None,
Epilog -> {Red, PointSize[0.02], Point[{3 Pi, 3 Pi}]}]


You can improve quality with plot points.

-

Define the function you want to plot f[x,y] and the value(s) you want to plot and then use the ContourPlot command. You get somewhat different pictures by choosing different regions of x,y to plot over.

f[x_, y_] := x Sin[y/x] + y Sin[2 x];
ContourPlot[f[x, y] == f[3 Pi, 3 Pi], {x, 0, 25}, {y, 0, 25}]


-
Awesome, thanks bill s! –  Logan Nov 10 '13 at 5:32
@ ubpdqn very sexy! –  Logan Nov 11 '13 at 21:56